

## **General Disclaimer**

### **One or more of the Following Statements may affect this Document**

- This document has been reproduced from the best copy furnished by the organizational source. It is being released in the interest of making available as much information as possible.
- This document may contain data, which exceeds the sheet parameters. It was furnished in this condition by the organizational source and is the best copy available.
- This document may contain tone-on-tone or color graphs, charts and/or pictures, which have been reproduced in black and white.
- This document is paginated as submitted by the original source.
- Portions of this document are not fully legible due to the historical nature of some of the material. However, it is the best reproduction available from the original submission.



SCHOOL OF ENGINEERING  
OLD DOMINION UNIVERSITY  
NORFOLK, VIRGINIA

Technical Report 76-T8

(NASA-CR-147145) RETRIEVAL OF SURFACE  
TEMPERATURE BY REMOTE SENSING (Old Dominion  
Univ. Research Foundation) 42 p HC \$4.00

CSCI 04A

N76-22637

Unclas

G3/43 26754

## RETRIEVAL OF SURFACE TEMPERATURE BY REMOTE SENSING

*By*

S.K. Gupta

*and*

S.N. Tiwari

*Prepared for the*  
National Aeronautics and Space Administration  
Langley Research Center  
Hampton, Virginia

*Under*  
Grant NSG 1153



April 1976

SCHOOL OF ENGINEERING  
OLD DOMINION UNIVERSITY  
NORFOLK, VIRGINIA

Technical Report 76-T8

## RETRIEVAL OF SURFACE TEMPERATURE BY REMOTE SENSING

*By*

S.K. Gupta

*and*

S.N. Tiwari

*Prepared for the*

National Aeronautics and Space Administration  
Langley Research Center  
Hampton, Virginia 23665

*Under*

Grant NSG 1153

Dr. Henry G. Reichle, Technical Monitor  
Environmental and Space Sciences Division

*Submitted by the*

Old Dominion University Research Foundation  
Norfolk, Virginia 23508



April 1976

## FOREWORD

This report constitutes a part of the work done on the research project entitled "Radiative Transfer Models for Nonhomogeneous Atmosphere." The work was supported by the NASA Langley Research Center through Grant No. NSG-1153. The grant was monitored by Dr. Henry G. Reichle.

# TABLE OF CONTENTS

	Page
FOREWORD . . . . .	ii
TABLE OF CONTENTS . . . . .	iii
LIST OF FIGURES . . . . .	iv
SUMMARY . . . . .	1
1. INTRODUCTION . . . . .	2
2. RETRIEVAL OF SURFACE TEMPERATURE . . . . .	4
3. SENSITIVITY CALCULATIONS . . . . .	10
3.1 Model Atmosphere . . . . .	10
3.2 Effect of CO Concentration . . . . .	12
3.3 Effect of Water Vapor Concentration . . . . .	14
3.4 Effect of Atmospheric Temperature Profile . . . . .	14
3.5 Effect of Surface Emittance . . . . .	17
3.6 Effect of Altitude of Observation . . . . .	17
4. CONCLUSIONS . . . . .	22
REFERENCES . . . . .	23
APPENDIX A - EXPLANATION OF SYMBOLS USED IN THE COMPUTER	
PROGRAM "SURFACE" . . . . .	24
APPENDIX B - PROGRAM "SURFACE" FOR TEMPERATURE	
RETRIEVAL . . . . .	28
B-1 Subroutine BROWN for Program SURFACE . . . . .	37

## LIST OF FIGURES

FIGURE NO.	TITLE
2.1	Variation of the integrated upwelling radiance at 10,500 ft. with surface temperature.
3.1	Effect of uncertainty in the assumed value of CO concentration on the retrieved values of surface temperature.
3.2	Effect of the uncertainty in the assumed water vapor profile on the retrieved values of surface temperature.
3.3	Effect of the uncertainty in the assumed temperature profile on the retrieved values of surface temperature.
3.4	Effect of uncertainty in the assumed value of surface emittance on the retrieved values of surface temperature.
3.5	Effect of uncertainty in the altitude of observation on the retrieved values of surface temperature.

RETRIEVAL OF SURFACE TEMPERATURE  
BY REMOTE SENSING

by

S. K. Gupta and S. N. Tiwari

School of Engineering  
Old Dominion University  
Norfolk, Virginia 23508

SUMMARY

A simple procedure and computer program have been developed for retrieving the surface temperature from the measurement of upwelling radiance in a single spectral region. The program evaluates the total upwelling radiance at any altitude in the region of the CO fundamental band ( $2070\text{--}2220\text{ cm}^{-1}$ ) for several values of surface temperature. Actual surface temperature is inferred by interpolation of the measured upwelling radiance between the computed values of radiance for the same altitude. Sensitivity calculations have been made to determine the effect of uncertainty in various surface, atmospheric and experimental parameters on the inferred value of surface temperature.

## 1. INTRODUCTION

The work presented in this report is a part of an ongoing effort to develop necessary procedures for the interpretation of the data obtained from a gas-filter correlation spectrometer. This instrument has been developed and is frequently flown for measuring carbon monoxide concentration in the troposphere. Like most of the passive remote sensing devices used for similar purposes, this instrument measures the upwelling radiation in the region of the CO fundamental band ( $2070\text{--}2220\text{ cm}^{-1}$ ). This upwelling radiation originates at the surface and is modified by the emission and absorption by various pollutants and other constituents in the atmosphere. The total radiant energy emitted by the earth's surface depends on both the emittance and the temperature of the surface. Reliable estimates of surface emittance, however, can be obtained more easily than those for the temperature. Therefore, the surface temperature is an important unknown parameter in the analysis of such radiation measurements.

Physical contact measurements of surface temperature over large areas of land or water (as are usually covered in remote sensing experiments) will be extremely time consuming and expensive, if at all possible. Further, it is important that the effective brightness temperature be determined for the particular spectral region in which the measured band is located. This temperature, in some cases, may be substantially different than that obtained in a contact measurement or from remote measurement in any other spectral region. This study is designed to examine the feasibility of obtaining the effective surface temperature from a radiometric measurement of the total upwelling radiance in the same band. Sensitivity calculations are performed to examine the effect of uncertainties in other relevant input parameters on the inferred value of the surface temperature.



The theoretical formulation of the transmittance models (line-by-line and quasi-random band model) and the computational procedures for evaluating the transmittance and upwelling atmospheric radiance are discussed in detail in [1,2]\*. For selected infrared bands of different gases, homogeneous path transmittances were calculated by employing the line-by-line and quasi-random band models in [2], and these were compared with the experimental results of Burch et al. [3]. The comparison of results indicated the existence of significant differences between the band model and experimental results. In view of the high accuracy requirement for atmospheric work, the line-by-line formulation was selected for use throughout the present investigation. A detailed description of the computational procedure and the computer program is given in [1].

The procedure for evaluating the upwelling radiance and retrieving the surface temperature is outlined in Sec. 2. Brief description of the important features of the computer program is also given in this section. The sensitivity calculations are performed to study the effect of uncertainties in CO concentration, water vapor density, atmospheric temperature profiles, surface emittance, and the altitude of observation on the inferred value of surface temperature. These are presented in Sec. 3.

---

\*The numbers in brackets indicate references.

## 2. RETRIEVAL OF SURFACE TEMPERATURE

The total upwelling radiance emergent from a plane-parallel atmosphere, in the absence of any significant scattering effects may be expressed as [1,4]

$$E(\omega) = E_G(\omega) + E_R(\omega) , \quad (2.1)$$

where  $E_G(\omega)$  is the thermal radiation emitted by the underlying surface and the atmosphere and  $E_R(\omega)$  is the incident solar radiation reflected by the surface.

The expression for the thermal radiation may be written as

$$E_G(\omega) = \epsilon B(\omega, T_s) \tau(\omega, 0) + \int_0^h B(\omega, T(z)) [d\tau(\omega, z)/dz] dz , \quad (2.2)$$

where  $\epsilon$  is the surface emittance and is assumed to be independent of frequency,  $B(\omega, T)$  is the Planck's blackbody function,  $T_s$  is the surface temperature,  $T(z)$  is the temperature at the altitude  $z$ , and  $\tau(\omega, z)$  is the monochromatic transmittance of the atmosphere between the top of the atmosphere and altitude  $z$ . The first term on the right hand side represents the radiation from the surface while the second term represents the radiation from the atmosphere. The reflected solar radiation component may be expressed as [1,4]

$$E_R(\omega) = (1/\pi)(1-\epsilon) \cos \theta H_S(\omega) [\tau(\omega)]^\zeta , \quad (2.3)$$

where  $\theta$  is the sun zenith angle,  $H_S(\omega)$  is the solar irradiance at the top of the atmosphere,  $\tau(\omega)$  is the transmittance vertically through the entire atmosphere.  $\zeta = 1 + f(\theta)$  where  $f(\theta) = \sec \theta$  for  $\theta \leq 60^\circ$  and

$f(\theta) = Ch(\theta)$  for  $\theta > 60^\circ$  and  $Ch(\theta)$  is the Chapman function. The Total upwelling radiance can be obtained by integrating the appropriate expressions over the frequency interval of the band.

It is clear from Eqs.(2.1)-(2.3) that the total upwelling radiance is a function of the surface emittance and temperature, atmospheric transmittance and temperature profile and sun zenith angle. If the underlying surface has a high emittance ( $\epsilon > 0.9$ ), it can be shown that in the region of the CO fundamental band, reflected solar radiation makes an insignificant contribution to the total upwelling radiance [4]. Table 2.1 shows a comparison of the total upwelling radiance to its surface component for seven values of surface temperature computed for an altitude of 10,500 ft. It shows that the surface component always constitutes a predominant part of the total radiance accounting for approximately 70 per cent for  $T_s = 290^\circ\text{K}$  to about 85 per cent for  $T_s = 320^\circ\text{K}$ . Figure 2.1 shows the dependence of the total upwelling radiance on the surface temperature. It has also been shown in the earlier report [1] that the dependence of the total upwelling radiance on the surface temperature is much greater than the small linear increase with the surface emittance.

The atmospheric component of the total radiance depends on the atmospheric temperature, pressure and water vapor profiles and the concentrations of the infrared active constituents like  $\text{N}_2\text{O}$ ,  $\text{CO}_2$ ,  $\text{CO}$ . Reliable profiles of atmospheric temperature, pressure and water vapor density are available routinely from radiosonde measurements and can be used in the present work. Average concentrations of  $\text{CO}_2$  and  $\text{N}_2\text{O}$  are known [5] with reasonable accuracy and can be used in conjunction with the above information. Extensive tabulation is available relating the nature and appearance of various surfaces

Table 2.1

Comparison of the Total Upwelling  
Radiance with its Surface Component  
for an Altitude of 10,500 ft.

Surface Temp. (°K)	Total Upwelling Radiance ( $10^{-5}$ watts $\text{cm}^{-2}\text{Sr}^{-1}$ )	Surface Component of Radiance ( $10^{-5}$ watts $\text{cm}^{-2}\text{Sr}^{-1}$ )
290	1.7141	1.2549
295	1.9621	1.5029
300	2.2484	1.7892
305	2.5771	2.1179
310	2.9526	2.4934
315	3.3795	2.9203
320	3.8627	3.4034




Fig. 2.1 Variation of the integrated upwelling radiance at 10,500 ft. with surface temperature.

to their emittance [6] and, therefore, reasonable estimate of the emittance of a surface can be made just from its appearance and some knowledge of its composition. It is possible to use Eqs.(2.1)-(2.3), along with the above information, to infer the temperature of the underlying surface in the following manner.

Total upwelling radiance is evaluated using Eqs.(2.1)-(2.3) at the altitude of observation for seven different values of the surface temperature ranging from 290°K to 320°K. Actual surface temperature is obtained by quadratic interpolation of the observed value of the upwelling radiance between the tabulated values of the computed radiances.

Since actual radiometric observations in this band are not available at present, the input values of the upwelling radiance (to be used as the actual radiometric data) are also generated theoretically. Upwelling radiance is computed for six different values of surface temperature using the same model atmosphere and surface emittance as adopted for the final retrieval procedure. Table 2.2 shows the surface temperatures used as input data in this program as well as those recovered by the retrieval program (named SURFACE) from the computed values of the upwelling radiances. The small differences between the input and retrieved temperatures (less than 0.1 °K) is representative of the errors introduced during the interpolation process. It is important to note that, because of the non-availability of actual radiometric observations to work with, the only useful purpose served by the above numerical exercise is to illustrate in principle that surface temperature can be inferred using this procedure. It also enables us to study the effect of the uncertainties in various other surface, atmospheric and experimental parameters on the inferred value of surface temperature as described in the next section.

Table 2.2

Input Surface Temperatures and  
Inferred Surface Temperatures  
from Computed Upwelling Radiances

Input Temp. (°K)	Inferred Temp. (°K)
292.50	292.41
297.50	297.41
302.50	302.41
307.50	307.41
312.50	312.42
317.50	317.42

The procedure for inferring the surface temperature from the V channel output of the gas-filter correlation instrument is under investigation at present.

### 3. SENSITIVITY CALCULATIONS

In addition to the surface temperature, several other atmospheric, surface, and experimental parameters have considerable effect on the total upwelling radiance. As discussed in the previous section, precise knowledge of these parameters is of vital importance in retrieving the surface temperature. Uncertainties in the values of these parameters will inevitably affect the inferred value of surface temperature. The extent of such effect is investigated in this section. The parameters whose influence is considered for the present study are, CO concentration, water vapor distribution, atmospheric temperature profile, surface emittance, and the altitude of observation.

The model atmosphere discussed in the next subsection was used to generate the input data (upwelling radiances to be used as radiometric observations in the retrieval program). Several values of the various parameters mentioned earlier were chosen within realistic ranges and used for data reduction to examine the sensitivity of the retrieved surface temperature to uncertainties in these parameters.

#### 3.1 Model Atmosphere

A model atmosphere has been chosen from a few different sources to form the basis of this sensitivity study and is given in Table 3.1. Information of this subsection (along with Table 3.1) form the base model for sensitivity calculations.

The atmosphere up to 17,500 ft. has been divided in 10 layers. Average temperature and pressure for each layer have been obtained from the U.S. Standard Atmosphere, 1962 [5], using interpolation where necessary. The standard water vapor profile, also shown in Table 3.1, has been taken from



McClatchey et al. [7] and average concentrations are computed again by interpolation. Carbon dioxide and nitrous oxide are assumed to be uniformly mixed in the atmosphere. For these gases, average value of concentrations are obtained from [7] as  $\text{CO}_2 = 330 \text{ ppmV}$ , and  $\text{N}_2\text{O} = 0.28 \text{ ppmV}$ . A CO concentration of  $0.2 \text{ ppmV}$ , which is typical of a polluted atmosphere up to 2-3 km, has been used for present work. A surface emittance value of 1.0 is used in some calculations and 0.9 in others. Throughout this investigation, the altitude of observation is assumed to be at 10,500 ft.

The altitudes at the layer boundaries are listed in the first column of Table 3.1. Data in the next three columns represent the averages for layers between the altitudes in first column appearing just above and below the values.

Table 3.1

Model Atmosphere used as the Basis  
of the Sensitivity Study

Altitude (ft)	Pressure (mm Hg)	Temperature (°K)	Water Vapor Conc. ( $10^3 \text{ ppmV}$ )
0			
500	763.16	287.66	7.626
1,500	742.65	286.17	7.240
2,500	716.03	284.19	6.726
4,500	677.55	281.22	5.972
6,500	628.87	277.26	5.100
8,500	583.07	273.30	4.222
10,500	540.03	269.34	3.338
12,500	499.59	265.38	2.669
14,500	461.66	261.42	2.074
17,500	417.57	256.48	1.493

With the information of this subsection as a base model, various results were obtained to study the influence of uncertainties in different parameters on the retrieved value of temperature. The results are presented in the following subsections. For a quantitative comparison, the retrieved surface temperatures, obtained with different values of the various parameters (and for an input surface temperature of 300 °K) are listed in Table 3.2. The last column of this table gives the difference in the retrieved temperatures obtained by varying the different parameters and that obtained by using the base model.

### 3.2 Effect of CO Concentration

To establish the effect of uncertainties in the CO concentration (in the atmosphere) on the retrieved surface temperatures, results were obtained for three different values of CO concentration, 0.1, 0.2, and 0.3 ppmV. These are illustrated in Fig. 3.1 where the solid curve represents the results for the base model (i.e., for the CO concentration of 0.2 ppmV). The value of surface emittance used in obtaining these results was unity.

From the figure it is noted that the uncertainty in the CO concentration has a small influence on the retrieved temperature values. As an example, consider the results obtained for the input surface temperature of 300 °K (see Table 3.2). For this case, the retrieved values of temperature obtained for the three CO concentrations of 0.1, 0.2, and 0.3 ppmV are 299.51, 299.91, and 300.24 °K respectively. This indicates that the retrieved values of surface temperature are not highly sensitive to the variation in CO concentration in the atmosphere.




Fig. 3.1 Effect of uncertainty in the assumed value of CO concentration on the retrieved values of surface temperature.

### 3.3 Effect of Water Vapor Concentration

The effect of uncertainties in the water vapor density profile on the retrieved values of surface temperature is shown in Fig 3.2. The solid line represents the results obtained with the standard water vapor profile while the dashed lines represent those obtained with half-standard and twice-standard water vapor profiles. The value of surface emittance used in this computation was also unity.

It can be seen from Fig. 3.2 that the effect of variations of the water vapor distribution on the retrieved surface temperature is considerably large. As shown in Table 3.2, for an input surface temperature of  $300^{\circ}\text{K}$  the retrieved values of surface temperature for half-standard, standard and twice-standard water vapor profiles are 299.07, 299.91, and  $301.21^{\circ}\text{K}$  respectively.

### 3.4 Effect of Atmospheric Temperature Profile

The effect of uncertainties in the atmospheric temperature profile on the retrieved values of surface temperature is illustrated in Fig. 3.3. The solid line represents the results obtained using the standard atmospheric temperature profile (as given in Table 3.1) while the dashed lines represent the results obtained with temperature profiles which are  $2^{\circ}\text{K}$  lower and  $2^{\circ}\text{K}$  higher than the standard values. Value of surface emittance used in this computation was again unity.

It can be seen from this figure that the effect of uncertainty in the temperature profile is significant but smaller than that for some other factors. Retrieved surface temperature values for an input temperature of  $300^{\circ}\text{K}$  for  $2^{\circ}\text{K}$  lower, standard and  $2^{\circ}\text{K}$  higher temperature profiles are 300.23, 299.91, and  $299.57^{\circ}\text{K}$  respectively (see Table 3.2).




Fig. 3.2 Effect of the uncertainty in the assumed water vapor profile on the retrieved values of surface temperature.




Fig. 3.3 Effect of the uncertainty in the assumed temperature profile on the retrieved values of surface temperature.

### 3.5 Effect of Surface Emittance

The effect of uncertainty in the assumed value of surface emittance on the retrieved values of surface temperature is shown in Fig. 3.4. The surface emittance value used with the model atmosphere for input data generation is 0.9. The solid line represents the retrieved values of surface temperature using  $\epsilon = 0.9$  while the dashed lines represent the results obtained by using  $\epsilon = 0.8$  and 1.0.

It can be seen from Fig. 3.4 that the uncertainty in the value of surface emittance has considerable effect on the retrieved value of surface temperature. For an input surface temperature of 300°K, retrieved values of surface temperature obtained with surface emittances of 0.8, 0.9, and 1.0 are 301.21, 299.91, and 298.88°K respectively. These results are also given in Table 3.2.

### 3.6 Effect of the Altitude of Observation

The altitude of observation (i.e., altitude of the flying aircraft) is a measured experimental parameter and is also subject to the usual uncertainties. The effect of the uncertainty in the aircraft altitude on the retrieved surface temperature values is shown in Fig. 3.5. An altitude of 10,500 ft. was used in the program which computed the input radiances. The solid line represents the results obtained by using 10,500 ft. as the altitude of observation while the dashed lines represent the results obtained with the altitudes of observation assumed at 10,000 and 11,000 ft. respectively. A surface emittance of 0.9 was used in obtaining these results.

It can be seen from Fig. 3.5 that the uncertainty in the altitude of observation has very little effect on the retrieved values of surface temperature. For the input surface temperature of 300°K, the retrieved values




Fig. 3.4 Effect of uncertainty in the assumed value of surface emittance on the retrieved values of surface temperature.






Fig. 3.5 Effect of uncertainty in the altitude of observation on the retrieved values of surface temperature.

of surface temperature for altitudes of observation at 10,000, 10,500, and 11,000 ft. are 299.74, 299.91, and 300.09°K respectively. These results are also given in Table 3.2.

Table 3.2

Retrieved Surface Temperature for  
Input Surface Temperature of 300°K  
with Uncertainty in Other Parameters

Parameters	Values of the Parameter	Retrieved Surface Temp. (°K)	Difference with Results from Base Model
CO Conc. (ppmV)	0.1	299.51	- 0.40
	0.2	299.91	-
	0.3	300.24	+ 0.33
H <sub>2</sub> O Conc.	1/2 Standard	299.07	- 0.84
	Standard	299.91	-
	2 x Standard	301.21	+ 1.30
Temperature Profile	Standard -2°K	300.23	+ 0.32
	Standard	299.91	-
	Standard +2°K	299.57	- 0.34
Surface Emittance	0.8	301.21	+ 1.30
	0.9	299.91	-
	1.0	298.88	- 1.03
Altitude of Observation (ft.)	10,000	299.74	- 0.17
	10,500	299.91	-
	11,000	300.09	+ 0.18

As indicated earlier, the fourth column in Table 3.2 lists the differences observed in the retrieved surface temperatures corresponding to the variation of the parameters listed in column 2. The middle value of each parameter refers to the standard atmosphere and the value of the retrieved surface temperature obtained for this case forms the basis of comparison. An inspection of Table 3.2 shows that the uncertainties in water vapor concentration and surface emittance are the most important factors affecting the accuracy of the retrieved values of surface temperature. Uncertainty in the CO concentration and the atmospheric temperature profile are the next in order of decreasing importance. The uncertainty in the altitude of observation is of least significance.

#### 4. CONCLUSIONS

This study reveals that it is possible to infer the temperature of the underlying surface from the measurement of upwelling infrared radiance. A computer program (named SURFACE) was developed specifically for this purpose. The sensitivity of the inferred surface temperature to the uncertainties in the values of the various parameters (CO concentration, water vapor concentration profile, atmospheric temperature profile, surface emittance, and altitude of observation) was investigated. It is found that the uncertainties in water vapor concentration and surface emittance are the most important factors affecting the accuracy of the inferred value of surface temperature. The other factors, in order of decreasing significance, are the CO concentration, atmospheric temperature profile, and the altitude of observation.

A new instrument (provided by TRW) is anticipated to be operational at the Langley Research Center in the near future. This instrument will have a radiometer channel which will operate in the region of the CO fundamental band. The results of the sensitivity study reported here would be directly applicable to the output of this radiometer channel.

## REFERENCES

1. S. K. Gupta and S. N. Tiwari, "Evaluation of Upwelling Infrared Radiance from Earth's Atmosphere," Technical Report 75-T14, School of Engineering, Old Dominion University, Norfolk, Virginia (1975).
2. S. K. Gupta and S. N. Tiwari, "Evaluation of Transmittance of Selected Infrared Bands," Technical Report 76-T7, School of Engineering, Old Dominion University, Norfolk, Virginia (1976).
3. D. E. Burch, D. Gryvnak, E. B. Singleton, W. L. France, and D. Williams, "Infrared Absorption by Carbon Dioxide, Water Vapor and Minor Atmospheric Constituents," AFCRL-62-698, Air Force Cambridge Research Laboratories, Bedford, Mass. (1962).
4. C. B. Ludwig, M. Griggs, W. Malkmus, and E. R. Bartle, "Air Pollution Measurements from Satellites," NASA CR-2324 (November 1973).
5. U. S. Standard Atmosphere, 1962, U. S. Gov't. Printing Office, Washington, D. C.
6. K. Ya. Kondratyev, "Radiation Characteristics of the Atmosphere and the Earth's Surface," NASA TTF-678, Amerind Publishing Co., New Delhi, India, (1973).
7. R. A. McClatchey, R. W. Fenn, J.E.A. Selby, F. E. Volz, and J. S. Garing, "Optical Properties of the Atmosphere (Third Edition)," AFCRL-72-0497, Air Force Cambridge Research Laboratories, Bedford, Mass. (1972).

## APPENDIX A

## EXPLANATION OF SYMBOLS USED IN THE COMPUTER PROGRAM "SURFACE"

AC	Array of monochromatic total absorption coefficients within one super-interval.
ACB	Absorption coefficient due to wing contributions at the lower frequency boundary of a super-interval.
ACE	Same as ACB, at the upper frequency boundary.
ACM	Same as ACB, at the center of the super-interval.
AL	Array of the half-widths of the individual lines.
ALFT	Altitude of observation in feet.
ALKM	Altitude of observation converted to kilometers.
ALX	Width of the narrow subintervals near the line center.
COMP	Atmospheric component of the upwelling radiance in one super-interval
DEL	Width of one super-interval.
DLIM	Width of the region from which the direct contribution is obtained.
EL	Array of the energies of the lower states for individual lines.
EMI	Surface emittance.
FACT	Factor used for obtaining the line-intensities corresponding to the temperature of each layer.
FGR	Gradients used in computing the filter function for each super-interval from the tabulated values.
FIL	Array of filter function values for all super-intervals.
FR	Array of frequencies for the individual lines.
FRB	Frequencies at the boundaries of the super-intervals.
FRC	Frequencies at the centers of the super-intervals.
FRD	Frequencies at the interval boundaries within one super-interval.

FRE	Frequencies of the lines within one interval.
FRG	Frequencies of the locations where absorption coefficients are evaluated within one super-interval.
FRL	Lower frequency boundary of the entire frequency range.
FRS	Frequencies at the sub-interval boundaries within one interval.
FRT	Tabulated values of the filter function.
FRU	Upper frequency boundary of the entire frequency range.
GR	Gradients used in computing solar irradiance at the top of the atmosphere in each super-interval.
HL	Computed value of the solar irradiance at the top of the atmosphere in each super-interval.
HS	Tabulated values of the solar irradiance at the top of the atmosphere at a few frequencies.
ID	Integer identifying the molecule of the origin of each line.
IER	Error parameter of the interpolation subroutine.
NG1,NG2 } NG3,NG4 }	Identifying integers for the absorbing gases being considered.
PART	A lumped constant taking account of the variation of rotational and vibrational partition functions.
PL	Optical path length from the altitude of observation up to the layer in question.
PLK	Planck's function for one layer in each interval.
PNT	Pressure at NTP (760 mm Hg)
PREC	Average pressure for each layer.
PSK	Planck's function for the surface in each interval at different surface temperatures.
QV	Concentrations of different absorbers (in ppmV) in each layer.

RADNC	Total upwelling radiance for one surface temperature in a super-interval.
RADNCE	Integrated upwelling radiance for the entire band for one surface temperature.
RCOM	Reflected solar radiation component of the upwelling radiance in one interval.
RDNC	Upwelling radiance for one surface temperature in one interval.
RP	Exponent which accounts for the temperature dependence of the rotational partition function.
SI	Array of the intensities of the individual lines.
SPTR	Pressure path lengths for different gases in each layer.
TEMC	Average temperature for each layer (in °K).
TEMS	Surface temperatures (in °K).
TEMSU	Retrieved surface temperatures (in °K).
THC	Thickness of each layer (in km).
TP	Total path length for the solar radiation from the top of the atmosphere to the instrument via ground reflection.
TR	Monochromatic transmittance at each of the frequencies FRG.
TRA	Transmittances averaged over each interval between altitude of observation and various layers.
TRT	Transmittance for the solar radiation averaged over each interval.
VALUE	Observed values of the upwelling radiance over the entire band.
VPF	Factor accounting for the temperature dependence of the vibrational partition function.
WF	Frequencies at which the filter function is tabulated.
WIDF	Factor used to convert line half-widths from reference values to those appropriate for each layer.
WN	Frequencies at which solar irradiance at the top of the atmosphere is tabulated.



WLIM	Width of the region from which the wing contribution is obtained.
WL1,WL2	Weighting factors used in the Gauss-Legendre quadrature formula.
XA	Array of some values of RADNCE used in the interpolation routine IUNI.
XL1,XL2	Abscissa values used in the Gauss-Legendre quadrature formula.
YA	Array of some values of TEMS used in the interpolation routine IUNI.
ZEN	Sun zenith angle.

## APPENDIX B

## PROGRAM SURFACE FOR TEMPERATURE RETRIEVAL

```

PROGRAM SURFACE(INPUT,OUTPUT)
  INTEGER X,Y,G,T
  DIMENSION VALUE(10),TEMSU(10),IER(10),XA(3),YA(3)
  DIMENSION PREC(10),TEMC(10),THC(10),QV(10,4),SPTR(10,4),VPF(10,4),
  /RP(4),GR(4),HS(5),WN(5),HL(150),FRT(31),WF(31),FIL(150),FRD(11),
  /FRE(20),FRS(26,10),PL(100,10),TR(100,10),TRA(11,10),PSK(7,10),
  /PLK(10,10),COMP(10),RDNC(7,10),TEMR(4),RADNC(7),RADNCE(7),TEMS(7),
  /FGR(30),TP(100,10),TRT(10)
  COMMON FR(1320),SI(1320),EL(1320),AL(1320),ID(1320),FRG(100,10),
  /AC(100,10),FRB(151),FRC(150),NP(10),WIDF(11,4),FACT(11,4),
  /PART(11,4),DLIM,WLIM,DELA,NGX,LE,K,G,L,PI
  READ 10, ALFT,EMI,ZEN
  READ 10, FRL,FRU,DEL,PNTPT,TNTP
  READ 11, NG,LE,NT,JT
  READ 10, DLIM,WLIM,ALX
  READ 10, (PREC(L),L=1,10)
  READ 10, (TEMC(L),L=1,10)
  READ 10, (TEMS(T),T=1,NT)
  READ 10, (TEMR(G),G=1,NG)
  READ 12, (THC(L),L=1,10)
  READ 12, (FRT(M),M=1,31)
  READ 12, ((VPF(L,G),L=1,10),G=1,NG)
  READ 13, XL1,XL2,WL1,WL2
  READ 14, ((QV(L,G),L=1,10),G=1,NG)
  READ 10, (RP(G),G=1,NG)
  READ 11, NG1,NG2,NG3,NG4
  READ 10, (WN(N),N=1,5)
  READ 14, (HS(N),N=1,5)
  READ 16, (VALUE(J),J=1,JT)
  READ 15, (FR(X),SI(X),EL(X),AL(X),ID(X),X=1,LE)
10  FORMAT(10F8.2)
11  FORMAT(16I5)
12  FORMAT(10F8.4)
13  FORMAT(8F10.6)
14  FORMAT(5E16.3)
15  FORMAT(2(F10.3,E12.4,F10.3,F5.3,I3))
16  FORMAT(6E12.4)
C   CONVERTS THE ALTITUDE OF OBSERVATION FROM FEET TO KILOMETERS AND
C   DETERMINES THE NUMBER OF LAYERS
    ALKM=ALFT*3.048E-04+0.1
    L=0
    STHC=0
100  L=L+1
    STHC=STHC+THC(L)

```

ORIGINAL PAGE IS  
OF POOR QUALITY

```

      ALR=ALKM-STHC
      IF (ALR.LT.0) GO TO 101
      LC=L
      GO TO 100
101  CONTINUE
      PI=3.14159
      LB=LC+1
C    CONVERTS WATER VAPOR CONCENTRATION FROM PPMV TO PR-CM
      DO 102 L=1,10
102  QV(L,1)=QV(L,1)/1245.
C    COMPUTES PATH-LENGTHS AND CONSTANTS REQUIRED TO CONVERT INTENSITIES
C    AND HALF-WIDTHS FROM REFERENCE TO AMBIENT CONDITIONS FOR ALL LAYERS
      CONST=0.1*TNTP/PNTP
      DO 103 L=1,10
      PTR=CONST*PREC(L)*THC(L)/TEMC(L)
      DO 103 G=1,NG
      SPTR(L,G)=PTR*QV(L,G)
      FACT(L,G)=1.439*(TEMC(L)-TEMR(G))/(TEMC(L)*TEMR(G))
      WIDF(L,G)=(SQRT(TEMR(G)/TEMC(L)))*PREC(L)/PNTP
103  PART(L,G)=VPF(L,G)*(TEMR(G)/TEMC(L))*RP(G)
C    COMPUTES THE NUMBER OF SUPER-INTERVALS AND THE FREQUENCIES AT THE
C    BOUNDARIES AND CENTERS OF THE SUPER-INTERVALS
      DELA=0.5*DEL
      RK=(FRU-FRL)/DEL+0.1
      KR=RK
      FRB(1)=FRL
      DO 104 K=1,KR
      FRC(K)=FRB(K)+DELA
104  FRB(K+1)=FRB(K)+DEL
      DELU=0.1*DEL
      ALY=2.*ALX
      ZEN=ZEN/57.29578
      CONS=18.*6.625*1.E-07
      CNST=6.625*0.3/1.38
C    COMPUTES SOLAR IRRADIANCE AT THE TOP OF THE ATMOSPHERE IN EACH
C    INTERVAL BY INTERPOLATION FROM THE TABULATED VALUES
      DO 105 N=1,4
105  GR(N)=(HS(N+1)-HS(N))/(WN(N+1)-WN(N))
      DO 106 K=1,KR
      N=0
107  N=N+1
      IF (FRC(K).LT.WN(N)) GO TO 106
      IF (FRC(K).GE.WN(N+1)) GO TO 107
      HL(K)=DEL*(HS(N)+GR(N)*(FRC(K)-WN(N)))

```

ORIGINAL PAGE IS  
OF POOR QUALITY

```

106 CONTINUE
C   COMPUTES FILTER FUNCTION FOR EACH INTERVAL FROM THE TABULATED VALUES
    WF(1)=FRL
    DO 152 M=1,30
        WF(M+1)=WF(M)+5.
152 FGR(M)=(FRT(M+1)-FRT(M))/(WF(M+1)-WF(M))
    DO 153 K=1,KR
        M=0
154 M=M+1
        IF (FRC(K).LT.WF(M)) GO TO 153
        IF (FRC(K).GE.WF(M+1)) GO TO 154
        FIL(K)=FRT(M)+FGR(M)*(FRC(K)-WF(M))
153 CONTINUE
C   INITIALIZES THE INTEGRATED UPWELLING RADIANCE FOR EACH SURFACE
C   TEMPERATURE
    DO 166 T=1,NT
166 RADNCE(T)=0.
C   THE ENTIRE OPERATION OF RADIANCE EVALUATION IS CARRIED OUT FOR
C   ONE SUPER-INTERVAL AT A TIME
    Y=0
    DO 109 K=1,KR
C   DIVIDES EACH SUPER-INTERVAL INTO 10 INTERVALS AND DEFINES
C   FREQUENCIES AT INTERVAL BOUNDARIES
        FRD(1)=FRB(K)
        DO 110 J=1,10
110 FRD(J+1)=FRD(J)+DELU
        DO 111 J=1,10
C   DETERMINES THE NUMBER OF LINES WITHIN EACH INTERVAL
            M=0
            MP=M
113 IF (Y.GE.LE) GO TO 112
            Y=Y+1
            IF (FR(Y).LT.FR(D(J))) GO TO 113
            IF (FR(Y).GE.FR(D(J+1))) GO TO 114
            M=M+1
            FRE(M)=FR(Y)
            MP=M
            GO TO 113
114 Y=Y-1
112 CONTINUE
C   CREATES SUB-INTERVALS WITHIN EACH INTERVAL AND UP-DATES THE
C   NUMBER OF SUB-INTERVALS NP EACH TIME
    N=1
    FRS(N,J)=FRD(J)

```

```

      IF (MP.LE.0) GO TO 115
      DO 116 M=1,MP
      DIF=FRE(M)-FRS(N,J)
      IF (DIF.LE.ALX) GO TO 117
      IF (DIF-ALY) 118,118,119
117  FRS(N+1,J)=FRE(M)
      N=N+1
      GO TO 120
118  FRS(N+1,J)=FRE(M)-ALX
      FRS(N+2,J)=FRE(M)
      N=N+2
      GO TO 120
119  FRS(N+1,J)=FRE(M)-ALY
      FRS(N+2,J)=FRE(M)-ALX
      FRS(N+3,J)=FRE(M)
      N=N+3
120  IF (M.EQ.MP) GO TO 121
      DIF=FRE(M+1)-FRS(N,J)
      IF (DIF.LE.ALX) GO TO 116
      IF (DIF.GT.ALY) GO TO 124
      FRS(N+1,J)=FRE(M)+ALX
      N=N+1
      GO TO 116
124  FRS(N+1,J)=FRE(M)+ALX
      FRS(N+2,J)=FRE(M)+ALY
      N=N+2
116  CONTINUE
121  DIF=FRD(J+1)-FRE(M)
      IF (DIF.LE.ALX) GO TO 125
      IF (DIF-ALY) 126,126,127
125  FRS(N+1,J)=FRD(J+1)
      NP(J)=N+1
      GO TO 111
126  FRS(N+1,J)=FRE(M)+ALX
      FRS(N+2,J)=FRD(J+1)
      NP(J)=N+2
      GO TO 111
127  FRS(N+1,J)=FRE(M)+ALX
      FRS(N+2,J)=FRE(M)+ALY
      FRS(N+3,J)=FRD(J+1)
      NP(J)=N+3
      GO TO 111
115  FRS(N+1,J)=FRD(J+1)
      NP(J)=N+1

```

```

111 CONTINUE
C   GENERATES THE FREQUENCY MESH BY COMPUTING FOUR FREQUENCY LOCATIONS
C   WITHIN EACH SUB-INTERVAL
      DO 128 J=1,10
        NQ=NP(J)-1
        DO 128 N=1,NQ
          VAR=0.5*(FRS(N+1,J)-FRS(N,J))
          CON=0.5*(FRS(N+1,J)+FRS(N,J))
          I=4*(N-1)+1
          FRG(I,J)=CON-VAR*XL1
          FRG(I+1,J)=CON-VAR*XL2
          FRG(I+2,J)=CON+VAR*XL2
128   FRG(I+3,J)=CON+VAR*XL1
C   INITIALIZES OPTICAL PATH-LENGTH AT EACH FREQUENCY LOCATION
      DO 129 J=1,10
        IG=4*(NP(J)-1)
        DO 129 I=1,IG
129   PL(I,J)=0.
C   STARTS COMPUTING OPTICAL PATH-LENGTHS FROM THE ALTITUDE OF
C   OBSERVATION TOWARDS THE GROUND INCLUDING LAYERS SUCCESSIVELY
      DO 130 M=1,LC
        L=LC+1-M
        IF (NG1.NE.2) GO TO 131
C   ADDS UP THE CONTRIBUTION OF WATER VAPOR
        G=1
        NGX=NG1
        CALL BROWN
        DO 132 J=1,10
          IG=4*(NP(J)-1)
          DO 132 I=1,IG
132   PL(I,J)=PL(I,J)+SPTR(L,G)*AC(I,J)
131   IF (NG2.NE.12) GO TO 133
C   ADDS UP THE CONTRIBUTION OF CARBON DIOXIDE
        G=2
        NGX=NG2
        CALL BROWN
        DO 134 J=1,10
          IG=4*(NP(J)-1)
          DO 134 I=1,IG
134   PL(I,J)=PL(I,J)+SPTR(L,G)*AC(I,J)
133   IF (NG3.NE.7) GO TO 135
C   ADDS UP THE CONTRIBUTION OF NITROUS OXIDE
        G=3
        NGX=NG3

```

ORIGINAL PAGE IS  
OF POOR QUALITY

```

      CALL BROWN
      DO 136 J=1,10
      IG=4*(NP(J)-1)
      DO 136 I=1,IG
136  PL(I,J)=PL(I,J)+SPTR(L,G)*AC(I,J)
135  IF (NG4.NE.1) GO TO 138
C    ADDS UP THE CONTRIBUTION OF CARBON MONOXIDE
      G=4
      NGX=NG4
      CALL BROWN
      DO 137 J=1,10
      IG=4*(NP(J)-1)
      DO 137 I=1,IG
137  PL(I,J)=PL(I,J)+SPTR(L,G)*AC(I,J)
138  CONTINUE
C    COMPUTES MONOCHROMATIC TRANSMITTANCES AT EACH MESH POINT
      DO 122 J=1,10
      IG=4*(NP(J)-1)
      DO 122 I=1,IG
      IF (PL(I,J).GT.675.) GO TO 123
      TR(I,J)=EXP(-PL(I,J))
      GO TO 122
123  TR(I,J)=0.
122  CONTINUE
C    COMPUTES TRANSMITTANCES AVERAGED OVER EACH INTERVAL
      DO 139 J=1,10
      NQ=NP(J)-1
      TRA(L,J)=0.
      DO 139 N=1,NQ
      VAR=0.5*(FRS(N+1,J)-FRS(N,J))
      I=4*(N-1)+1
      SUM1=TR(I,J)+TR(I+3,J)
      SUM2=TR(I+1,J)+TR(I+2,J)
      SUM=WL1*SUM1+WL2*SUM2
139  TRA(L,J)=TRA(L,J)+SUM*VAR/DELU
130  CONTINUE
C    STARTS EVALUATING THE TOTAL OPTICAL PATH-LENGTH FOR THE ENTIRE
C    ATMOSPHERE TO TAKE INTO ACCOUNT THE ATTENUATION OF SOLAR RADIATION
      DO 171 J=1,10
      IG=4*(NP(J)-1)
      DO 171 I=1,IG
171  TP(I,J)=PL(I,J)
      IF (LB.GT.10) GO TO 183
C    STARTS CONSIDERING THE LAYERS ABOVE THE ALTITUDE OF OBSERVATION

```

ORIGINAL PAGE IS  
OF POOR QUALITY

```

      DO 172 L=LB,10
      IF (NG1.NE.2) GO TO 173
C    ADDS UP THE CONTRIBUTION OF WATER VAPOR
      G=1
      NGX=NG1
      CALL BROWN
      DO 174 J=1,10
      IG=4*(NP(J)-1)
      DO 174 I=1,IG
174  TP(I,J)=TP(I,J)+SPTR(L,G)*AC(I,J)
173  IF (NG2.NE.12) GO TO 175
C    ADDS UP THE CONTRIBUTION OF CARBON DIOXIDE
      G=2
      NGX=NG2
      CALL BROWN
      DO 176 J=1,10
      IG=4*(NP(J)-1)
      DO 176 I=1,IG
176  TP(I,J)=TP(I,J)+SPTR(L,G)*AC(I,J)
175  IF (NG3.NE.7) GO TO 177
C    ADDS UP THE CONTRIBUTION OF NITROUS OXIDE
      G=3
      NGX=NG3
      CALL BROWN
      DO 178 J=1,10
      IG=4*(NP(J)-1)
      DO 178 I=1,IG
178  TP(I,J)=TP(I,J)+SPTR(L,G)*AC(I,J)
177  IF (NG4.NE.1) GO TO 172
C    ADDS UP THE CONTRIBUTION OF CARBON MONOXIDE
      G=4
      NGX=NG4
      CALL BROWN
      DO 179 J=1,10
      IG=4*(NP(J)-1)
      DO 179 I=1,IG
179  TP(I,J)=TP(I,J)+SPTR(L,G)*AC(I,J)
172  CONTINUE
183  CONTINUE
C    CONVERTS THE ABOVE PATH-LENGTH TO SLANT PATH TO TAKE INTO ACCOUNT
C    THE SUN ZENITH ANGLE AND ADDS TO IT THE PATH-LENGTH FROM GROUND
C    TO THE INSTRUMENT AND THEN COMPUTES MONOCHROMATIC TRANSMITTANCES
C    FOR THE ENTIRE PATH
      DO 180 J=1,10

```



```

      IG=4*(NP(J)-1)
      DO 180 I=1,IG
      TP(I,J)=TP(I,J)/ZEN+PL(I,J)
      IF (TP(I,J).GT.675.) GO TO 182
      TR(I,J)=EXP(-TP(I,J))
      GO TO 180
182  TR(I,J)=0.
180  CONTINUE
C    COMPUTES TRANSMITTANCES AVERAGED OVER EACH INTERVAL FOR THE
C    TOTAL PATH OF THE SOLAR RADIATION
      DO 181 J=1,10
      TRT(J)=0.
      NQ=NP(J)-1
      DO 181 N=1,NQ
      I=4*(N-1)+1
      VAR=0.5*(FRS(N+1,J)-FRS(N,J))
      SUM1=TR(I,J)+TR(I+3,J)
      SUM2=TR(I+1,J)+TR(I+2,J)
      SUM=WL1*SUM1+WL2*SUM2
181  TRT(J)=TRT(J)+SUM*VAR/DELU
      DO 140 J=1,10
140  TRA(LB,J)=1.
C    STARTS COMPUTING THE PLANCK FUNCTIONS
      PNUM=DELU*CONS*FRC(K)*FRC(K)*FRC(K)*1.E-07
      EEX=CNST*FRC(K)
C    COMPUTES PLANCK FUNCTIONS FOR THE SURFACE AT DIFFERENT
C    TEMPERATURES IN EACH INTERVAL
      DO 142 T=1,NT
      PS=PNUM/(EXP(EEX/TEMS(T))-1.)
      DO 142 J=1,10
142  PSK(T,J)=PS
C    COMPUTES PLANCK FUNCTIONS FOR DIFFERENT LAYERS IN EACH INTERVAL
      DO 143 L=1,LC
      PP=PNUM/(EXP(EEX/TEMC(L))-1.)
      DO 143 J=1,10
143  PLK(L,J)=PP
C    EVALUATES UPWELLING RADIANCE FOR DIFFERENT SURFACE TEMPERATURE
C    VALUES IN EACH INTERVAL
      DO 144 J=1,10
      RCOM=(1.-EMI)*DELU*ZEN*HL(K)*TRT(J)
      COMP(J)=0.
      DO 145 L=1,LC
145  COMP(J)=COMP(J)+PLK(L,J)*(TRA(L+1,J)-TRA(L,J))
      DO 146 T=1,NT

```

```

146 RDNC(T,J)=RCOM+COMP(J)+EMI*PSK(T,J)*TRA(1,J)
144 CONTINUE
C   INTEGRATES UPWELLING RADIANCE OVER ONE SUPER-INTERVAL
      DO 161 T=1,NT
        RADNC(T)=0.
        DO 161 J=1,10
161  RADNC(T)=RADNC(T)+RDNC(T,J)
C   CONVOLUTES THE UPWELLING RADIANCE WITH THE FILTER FUNCTION AND
C   INTEGRATES OVER THE FREQUENCY RANGE OF THE ENTIRE BAND
      DO 167 T=1,NT
167  RADNCE(T)=RADNCE(T)+RADNC(T)*FIL(K)
109 CONTINUE
C   EVALUATES THE ACTUAL SURFACE TEMPERATURE BY INTERPOLATING THE
C   OBSERVED UPWELLING RADIANCE BETWEEN THE UPWELLING RADIANCES
C   COMPUTED FOR DIFFERENT SURFACE TEMPERATURES
      NMAX=3
      NN=3
      NTAB=1
      DO 168 J=1,JT
        T=-1
169  T=T+2
        IF (VALUE(J).LT.RADNCE(T)) GO TO 168
        IF (VALUE(J).GE.RADNCE(T+2)) GO TO 169
        XO=VALUE(J)
        DO 170 N=1,NN
          XA(N)=RADNCE(T+N-1)
170  YA(N)=TEMS(T+N-1)
          IPT=-1
          CALL IUNI(NMAX,NN,XA,NTAB,YA,IORDER,XO,YO,IPT,IERR)
          TEMSU(J)=YO
          IER(J)=IERR
168 CONTINUE
      PRINT 60, (TEMS(T),RADNCE(T),T=1,NT)
60  FORMAT(1H1////(F10.2,E15.5,///))
      PRINT 62, (TEMSU(J),IER(J),J=1,JT)
62  FORMAT(1H1////(F15.2,I10,///))
      STOP
      END

```

B-1. Subroutine BROWN for Program SURFACE

```

SUBROUTINE BROWN
  INTEGER X,G
  COMMON FR(1320),SI(1320),EL(1320),AL(1320),ID(1320),FRG(100,10),
  /AC(100,10),FRB(151),FRC(150),NP(10),WIDF(11,4),FACT(11,4),
  /PART(11,4),DLIM,WLIM,DELA,NGX,LE,K,G,L,PI
C  INITIALIZES ABSORPTION COEFFICIENTS AT THE MESH POINTS FOR THE
C  DIRECT CONTRIBUTION AND AT THE SUPER-INTERVAL BOUNDARY AND
C  CENTER FOR THE WING CONTRIBUTION
    ACB=0.
    ACM=0.
    ACE=0.
    DO 300 J=1,10
      IG=4*(NP(J)-1)
      DO 300 I=1,IG
300 AC(I,J)=0.
C  STARTS EVALUATING THE CONTRIBUTION OF EACH LINE
    DO 301 X=1,LE
      IF (NGX.NE.ID(X)) GO TO 301
      DIF =ABS(FR(X)-FRC(K))
      IF (DIF.GT.WLIM) GO TO 301
      SIA=SI(X)*PART(L,G)*EXP(EL(X)*FACT(L,G))
      ALB=AL(X)*WIDF(L,G)
      IF (DIF.GT.DLIM) GO TO 302
C  EVALUATES THE DIRECT CONTRIBUTION AT EACH MESH POINT
      DO 303 J=1,10
        IG=4*(NP(J)-1)
        DO 303 I=1,IG
          FD= FR(X)-FRG(I,J)
          DEN=PI*(FD*FD+ALB*ALB)
303 AC(I,J)=AC(I,J)+SIA*ALB/DEN
          GO TO 301
C  EVALUATES WING CONTRIBUTION AT SUPER-INTERVAL BOUNDARIES AND CENTER
302 FB =FR(X)-FRB(K)
      FM =FR(X)-FRC(K)
      FE =FR(X)-FRB(K+1)
      PNUM=SIA*ALB
      ACB=ACB+PNUM/(PI*FB*FB)
      ACM=ACM+PNUM/(PI*FM*FM)
      ACE=ACE+PNUM/(PI*FE*FE)

```

```
301 CONTINUE
C   COMPUTES SLOPES FOR WING CONTRIBUTION BETWEEN SUPER-INTERVAL
C   BOUNDARIES AND CENTER
      SL1 =(ACM-ACB)/(FRC(K)-FRB(K))
      SL2 =(ACE-ACM)/(FRB(K+1)-FRC(K))
C   EVALUATES WING CONTRIBUTION AT EACH MESH POINT AND ADDS THE
C   DIRECT AND WING CONTRIBUTIONS
      DO 304 J=1,10
        IG=4*(NP(J)-1)
        DO 304 I=1,IG
          DIF=(FRG(I,J)-FRB(K))
          IF (DIF.GE.DELA) GO TO 305
          AC(I,J)=AC(I,J)+ACB+SL1*DIF
          GO TO 304
305 AC(I,J)=AC(I,J)+ACM+SL2*(DIF-DELA)
304 CONTINUE
      RETURN
      END
```